

Projet Pratique — Éléments Finis

Résolution du problème de gauchissement de Saint-Venant
par éléments triangulaires CST en déformation plane

Oumkalthoum M'HAMDI

Année universitaire 2025-2026

Contents

1	Introduction	2
2	Formulation du problème	2
2.1	Problème physique — Torsion de Saint-Venant	2
2.2	Équation forte	2
2.3	Formulation variationnelle (forme faible)	3
3	Discrétisation par éléments finis	3
3.1	Élément triangulaire CST — Fonctions de forme	3
3.2	Matrice B — Opérateur gradient discret	4
3.3	Matrice de rigidité élémentaire	4
3.4	Vecteur de charge élémentaire	4
3.5	Assemblage global	5
4	Implémentation Fortran	5
4.1	Structure du programme	5
4.2	Boucle d’assemblage	5
5	Résultats et interprétation	6
5.1	Maillage	6
5.2	Fonction de gauchissement ω	6
5.3	Profils de ω	7
5.4	Contraintes de cisaillement	7
6	Vérifications numériques	8
6.1	Comparaison de deux maillages et convergence qualitative	8
6.2	Vérification de l’équilibre global	9
6.3	Cohérence physique	9
7	Conclusion	9
A	Code source — TP_elas.f90	10

1 Introduction

Dans ce travail, la méthode des éléments finis (MEF) est utilisée pour étudier le problème de gauchissement de Saint-Venant d'une section soumise à un effort de torsion. Le problème se ramène à la résolution de l'équation de Laplace sur la section transversale Ω , avec des conditions aux limites de Neumann sur le contour $\partial\Omega$.

L'objectif est de mettre en place un solveur éléments finis permettant de résoudre ce problème de torsion. Plus précisément :

- Formuler le problème variationnel associé à l'équation de Laplace,
- Discrétiser le domaine par des éléments triangulaires linéaires (CST),
- Implémenter l'assemblage de la matrice de rigidité et du vecteur de charge en Fortran 90,
- Résoudre le système linéaire et post-traiter les résultats.

2 Formulation du problème

2.1 Problème physique — Torsion de Saint-Venant

On considère une poutre de section droite $\Omega \subset \mathbb{R}^2$, d'axe x , soumise à un moment de torsion M_t . D'après la théorie de Saint-Venant, le champ de déplacement s'écrit :

$$u_x = \alpha \omega(y, z), \quad u_y = -\alpha x z, \quad u_z = \alpha x y \quad (1)$$

où α est l'angle de torsion par unité de longueur et $\omega(y, z)$ est la **fonction de gauchissement**, inconnue scalaire définie sur la section transversale paramétrée par (y, z) . Les composantes de déformation non nulles sont :

$$\varepsilon_{xy} = \frac{\alpha}{2} \left(\frac{\partial \omega}{\partial y} - z \right), \quad \varepsilon_{xz} = \frac{\alpha}{2} \left(\frac{\partial \omega}{\partial z} + y \right) \quad (2)$$

et les contraintes de cisaillement correspondantes :

$$\tau_{xy} = G \alpha \left(\frac{\partial \omega}{\partial y} - z \right), \quad \tau_{xz} = G \alpha \left(\frac{\partial \omega}{\partial z} + y \right) \quad (3)$$

2.2 Équation forte

L'équilibre en l'absence de forces de volume conduit, après substitution du champ de contraintes, à l'équation de Laplace :

$$\frac{\partial^2 \omega}{\partial y^2} + \frac{\partial^2 \omega}{\partial z^2} = \Delta \omega = 0 \quad \text{dans } \Omega \quad (4)$$

La condition naturelle sur le contour $\partial\Omega$ (condition de surface libre : contrainte normale nulle $\sigma_{nn} = 0$) s'écrit :

$$\boxed{\begin{cases} \Delta \omega = 0 & \text{dans } \Omega \\ \frac{\partial \omega}{\partial n} = z n_y - y n_z & \text{sur } \partial\Omega \end{cases}} \quad (5)$$

où $\mathbf{n} = (n_y, n_z)$ est la normale extérieure unitaire au contour. Ce problème est de **Neumann pur** : si ω est solution, $\omega + C$ l'est aussi pour toute constante C . La condition de compatibilité de Neumann est vérifiée :

$$\oint_{\partial\Omega} \frac{\partial\omega}{\partial n} d\Gamma = \oint_{\partial\Omega} (z n_y - y n_z) d\Gamma = \int_{\Omega} \left(\frac{\partial z}{\partial y} - \frac{\partial y}{\partial z} \right) d\Omega = 0 \quad (6)$$

2.3 Formulation variationnelle (forme faible)

On multiplie l'équation forte par une fonction test $v \in H^1(\Omega)$ et on intègre sur Ω :

$$\int_{\Omega} v \Delta\omega d\Omega = 0 \quad (7)$$

En appliquant la formule de Green : $\int_{\Omega} v \Delta\omega d\Omega = - \int_{\Omega} \nabla v \cdot \nabla\omega d\Omega + \oint_{\partial\Omega} v \frac{\partial\omega}{\partial n} d\Gamma$, Après introduction de la condition de Neumann dans le terme de bord, on obtient :

$$\int_{\Omega} \nabla\omega \cdot \nabla v d\Omega = \oint_{\partial\Omega} v (z n_y - y n_z) d\Gamma \quad (8)$$

Le membre droit se transforme à nouveau par Green ($\oint_{\partial\Omega} v z n_y d\Gamma = \int_{\Omega} \partial(vz)/\partial y d\Omega = \int_{\Omega} z \partial v/\partial y d\Omega$), On arrive finalement à la forme faible suivante :

Trouver $\omega \in H^1(\Omega)$ tel que :

$$\boxed{\int_{\Omega} \nabla\omega \cdot \nabla v d\Omega = \int_{\Omega} \left(-z \frac{\partial v}{\partial y} + y \frac{\partial v}{\partial z} \right) d\Omega \quad \forall v \in H^1(\Omega)} \quad (9)$$

3 Discrétisation par éléments finis

3.1 Élément triangulaire CST — Fonctions de forme

Le domaine Ω est discrétisé en N_{el} éléments triangulaires à 3 nœuds (CST — *Constant Strain Triangle*). Sur chaque élément e de nœuds $(y_i, z_i)_{i=1,2,3}$, la fonction de gauchissement est approchée par un polynôme complet de degré 1 :

$$\omega^e(y, z) = a + b y + c z = \begin{pmatrix} 1 & y & z \end{pmatrix} \begin{pmatrix} a \\ b \\ c \end{pmatrix} \quad (10)$$

Les trois inconnues (a, b, c) sont déterminées en imposant l'interpolation aux nœuds $\omega^e(y_i, z_i) = \omega_i$:

$$\underbrace{\begin{pmatrix} 1 & y_1 & z_1 \\ 1 & y_2 & z_2 \\ 1 & y_3 & z_3 \end{pmatrix}}_M \begin{pmatrix} a \\ b \\ c \end{pmatrix} = \begin{pmatrix} \omega_1 \\ \omega_2 \\ \omega_3 \end{pmatrix} \Rightarrow \begin{pmatrix} a \\ b \\ c \end{pmatrix} = M^{-1} \begin{pmatrix} \omega_1 \\ \omega_2 \\ \omega_3 \end{pmatrix} \quad (11)$$

En substituant dans l'interpolation :

$$\omega^e(y, z) = \underbrace{\begin{pmatrix} 1 & y & z \end{pmatrix}}_{\mathbf{N}(y,z)=[N_1, N_2, N_3]} M^{-1} \begin{pmatrix} \omega_1 \\ \omega_2 \\ \omega_3 \end{pmatrix} = \mathbf{N}(y, z) \boldsymbol{\omega}^e \quad (12)$$

Les fonctions de forme N_i sont des polynômes linéaires en (y, z) vérifiant $N_i(y_j, z_j) = \delta_{ij}$. Leur expression analytique fait intervenir l'aire S^e de l'élément :

$$N_i(y, z) = \frac{1}{2S^e} [(y_j z_k - y_k z_j) + (z_j - z_k)y + (y_k - y_j)z] \quad (i, j, k) \text{ permutation circulaire de } (1, 2, 3) \quad (13)$$

avec $S^e = \frac{1}{2} |(y_2 - y_1)(z_3 - z_1) - (y_3 - y_1)(z_2 - z_1)|$.

3.2 Matrice B — Opérateur gradient discret

Le gradient de ω^e s'obtient en dérivant l'interpolation :

$$\nabla \omega^e = \begin{pmatrix} \partial \omega^e / \partial y \\ \partial \omega^e / \partial z \end{pmatrix} = \begin{pmatrix} b \\ c \end{pmatrix} = \underbrace{\begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}}_{\text{gradN}} M^{-1} \begin{pmatrix} \omega_1 \\ \omega_2 \\ \omega_3 \end{pmatrix} = B \boldsymbol{\omega}^e \quad (14)$$

La matrice gradN extrait les lignes 2 et 3 de M^{-1} , correspondant aux dérivées $\partial/\partial y$ et $\partial/\partial z$.

Propriété fondamentale du CST : puisque b et c sont des constantes (indépendantes de (y, z)), le gradient $\nabla \omega^e$ — et donc la déformation — est *uniforme sur tout l'élément*.

3.3 Matrice de rigidité élémentaire

En substituant $\omega_h = \mathbf{N} \boldsymbol{\omega}^e$ et $v_h = \mathbf{N} \mathbf{v}^e$ dans la forme faible (9), le membre gauche élémentaire donne :

$$\int_{\Omega^e} \nabla v_h \cdot \nabla \omega_h \, d\Omega = (\mathbf{v}^e)^T \underbrace{\int_{\Omega^e} B^T B \, d\Omega}_{K^e} \boldsymbol{\omega}^e \quad (15)$$

Puisque B est constant sur l'élément, l'intégration est immédiate :

$$\boxed{K^e = S^e B^T B} \in \mathbb{R}^{3 \times 3} \quad (16)$$

3.4 Vecteur de charge élémentaire

Le membre droit élémentaire de (9) vaut :

$$\int_{\Omega^e} \left(-z \frac{\partial v_h}{\partial y} + y \frac{\partial v_h}{\partial z} \right) d\Omega = (\mathbf{v}^e)^T \underbrace{\int_{\Omega^e} B^T \begin{pmatrix} -z \\ y \end{pmatrix} d\Omega}_{\mathbf{q}^e} \quad (17)$$

B étant constant, il reste à intégrer $[-z, y]^T$ sur Ω^e . On utilise la formule d'intégration exacte sur un triangle : pour une fonction linéaire $f(y, z) = \alpha + \beta y + \gamma z$, $\int_{\Omega^e} f \, d\Omega = S^e \cdot f(y_c, z_c)$ où $(y_c, z_c) = \frac{1}{3} \sum_i (y_i, z_i)$ est le **centroïde**. Ceci constitue le point de Gauss exact pour le CST.

On obtient :

$$\boxed{\mathbf{q}^e = S^e B^T \begin{pmatrix} -z_c \\ y_c \end{pmatrix}} \in \mathbb{R}^3 \quad (18)$$

3.5 Assemblage global

La matrice globale K et le vecteur $\mathbf{f}$ sont obtenus par **assemblage** : chaque contribution élémentaire est ajoutée aux indices globaux des nœuds de l'élément :

$$K_{[\text{nodelem}(r,e), \text{nodelem}(c,e)]} += K_{r,c}^e, \quad f_{[\text{nodelem}(r,e)]} += q_r^e \quad \forall r, c \in \{1, 2, 3\} \quad (19)$$

Le système global $K\omega = f$ est singulier, ce qui est classique pour un problème de Neumann pur. Pour assurer l'unicité, on impose $\omega = 0$ au nœud i_c le plus proche du centroïde géométrique du maillage :

$$K_{i_c, \cdot} = \mathbf{e}_{i_c}^T, \quad f_{i_c} = 0 \quad (20)$$

Cette condition permet simplement de lever l'indétermination liée à la constante additive. Le système est ensuite résolu par **factorisation LU** (routines LAPACK `dgetrf/dgetrs`).

4 Implémentation Fortran

4.1 Structure du programme

Le solveur est organisé en six étapes :

1. Lecture du maillage via le module `mesh.f90`
2. Allocation de $K \in \mathbb{R}^{N_{no} \times N_{no}}$ et $\mathbf{f} \in \mathbb{R}^{N_{no}}$
3. Boucle d'assemblage sur les N_{el} éléments
4. Imposition de la condition $\omega(i_c) = 0$
5. Résolution par factorisation LU (LAPACK : `dgetrf/dgetrs`)
6. Export au format Gmsh `resu.msh`

4.2 Boucle d'assemblage

Listing 1: Extrait de la boucle d'assemblage — `TP_elas.f90`

```

1  do i = 1, nelelem
2      ! Coordonnees recentrees des noeuds de l'element
3      y1 = coor(1,n1) ; z1 = coor(2,n1)
4      ...
5      ! Matrice de geometrie et son inverse
6      M_mat(1,:) = [1.0d0, y1, z1]
7      call invermat(3, M_mat, M_inv)
8      ! Matrice gradient : B = gradN * M^{-1}
9      B_mat = matmul(gradN, M_inv)
10     ! Contributions elementaires
11     Kel = selem * matmul(transpose(B_mat), B_mat)
12     qel = selem * matmul(transpose(B_mat), [-z_c, y_c])
13     ! Assemblage dans K et f globaux
14     Rsys(nodelem(r,i), nodelem(c,i)) += Kel(r,c)
15     vecs(nodelem(r,i), 1) += qel(r)
16 end do

```

5 Résultats et interprétation

5.1 Maillage

La section rectangulaire étudiée est de dimensions $y \in [-0.05, 0.05]$ m (direction courte) et $z \in [-0.4375, 0.4375]$ m (direction longue), soit un rapport d'élancement $L/e \approx 9$. Le maillage par défaut comporte $N_{no} = 316$ nœuds et $N_{el} = 532$ éléments triangulaires CST.




Figure 1: Discrétisation de la section rectangulaire par éléments triangulaires CST.

5.2 Fonction de gauchissement ω




Figure 2: Distribution 2D de la fonction de gauchissement $\omega(y, z)$. Les valeurs positives et négatives illustrent l'antisymétrie de la solution.




Figure 3: Vue 3D de $\omega(y, z)$. La forme en selle de cheval est caractéristique du gauchissement d'une section rectangulaire.

Les résultats numériques retrouvent bien les symétries attendues :

- **Antisymétrie en y** : $\omega(-y, z) = -\omega(y, z)$,
- **Antisymétrie en z** : $\omega(y, -z) = -\omega(y, z)$,
- **Extrema aux coins** : $|\omega|_{\max} \approx 0.0196 \text{ m}^2$.

Pour une section très élancée ($L/e \gg 1$), la solution analytique approche $\omega \approx -yz$, ce qui prédit des extrema aux coins valant $|y_{\max} \times z_{\max}| = 0.05 \times 0.4375 = 0.02188 \text{ m}^2$. La valeur FEM de 0.0196 m^2 est légèrement inférieure, ce qui s'explique par la correction apportée par les termes de la série de Saint-Venant qui atténuent le gauchissement près des bords courts. Même avec le maillage le plus grossier, la forme générale du champ de gauchissement est déjà correctement reproduite.

5.3 Profils de ω




Figure 4: Profils 1D de la fonction de gauchissement. (gauche) $\omega(0, z)$: profil selon z à $y \approx 0$, quasi-nul conformément à l'antisymétrie de la solution ; (droite) $\omega(y, 0)$: profil selon y à $z \approx 0$, valeurs de l'ordre de 10^{-5} m^2 (essentiellement zéro).

Les deux profils confirment que $\omega(0, z) \approx 0$ et $\omega(y, 0) \approx 0$, ce qui est cohérent avec l'antisymétrie théorique de la fonction de gauchissement pour une section rectangulaire centrée à l'origine. Le bruit résiduel sur le profil $\omega(0, z)$ est dû à la discrétisation : les nœuds sélectionnés à $y \approx 0$ ne sont pas exactement sur l'axe de symétrie.

5.4 Contraintes de cisaillement

À partir de ω , les contraintes de cisaillement sont dérivées par :

$$\tau_{xy} = G \alpha \left(\frac{\partial \omega}{\partial y} - z \right), \quad \tau_{xz} = G \alpha \left(\frac{\partial \omega}{\partial z} + y \right) \quad (21)$$




Figure 5: Distribution des contraintes de cisaillement sur la section.

Interprétation physique: Contrainte τ_z : Elle présente une variation quasi-linéaire selon z (terme dominant $-z$ dans τ_{xy}), avec des valeurs maximales aux bords supérieur et inférieur ($z = \pm z_{\max}$). Ce comportement est cohérent avec la théorie élémentaire de la torsion pour une section élancée.

Contrainte τ_y : Elle est quasi-nulle sur la majeure partie de la section et concentrée aux extrémités $z = \pm z_{\max}$ (bords longs). Pour une section $L/e \approx 9$, la torsion est essentiellement résistée par τ_z ; la contribution de τ_y est un effet de bord localisé.

Ce résultat est physiquement attendu : dans une section très élancée, le comportement se rapproche de celui d'une plaque mince en torsion, où seule la contrainte perpendiculaire à la grande dimension est significative.

6 Vérifications numériques

6.1 Comparaison de deux maillages et convergence qualitative

Quatre niveaux de raffinement ont été testés, de $N_{no} = 316$ nœuds (maillage 2D) à $N_{no} = 17\,417$ nœuds (maillage 5D). Le tableau 1 résume les extrema de ω pour chaque maillage.

Table 1: Extrema de ω en fonction du raffinement du maillage. Le maillage 5D sert de référence.

Maillage	N_{no}	N_{el}	$\omega_{\max}$ (m ²)	$ \omega_{\min} $ (m ²)
2D	316	532	0.019646	0.019593
3D	1163	2128	0.019636	0.019627
4D	4453	8512	0.019646	0.019640
5D	17417	34048	0.019651	0.019645




Figure 6: (gauche) Champ ω sur le maillage 5D ; (centre) carte de différence $\omega_{5D} - \omega_{2D}$; (droite) convergence des extrema de ω en fonction du nombre de nœuds.

Analyse: La différence relative maximale entre le maillage le plus grossier (2D) et le plus fin (5D) est de **0.96%**, ce qui confirme une convergence très rapide. La carte de différence montre que l'écart est distribué de façon quasi-uniforme sur toute la section, sans concentration particulière aux coins.

Ce comportement s'explique physiquement : pour une section très élancée ($L/e \approx 9$), la solution approche $\omega \approx -yz$, qui est une fonction *linéaire* en (y, z) . Or, les éléments CST ($\mathcal{P}_1$)

reproduisent exactement les fonctions linéaires, ce qui explique que même le maillage grossier capture l'essentiel de la solution.

On note également que $\omega_{\max}$ présente un comportement non-monotone (légère baisse de 2D à 3D), imputable à la légère asymétrie du maillage ($y \in [-0.0503, 0.0497]$ m), qui décale les nœuds extrémaux d'un raffinement à l'autre.

6.2 Vérification de l'équilibre global

Pour un problème de Neumann pur, la condition de compatibilité impose que l'intégrale du second membre soit nulle :

$$\int_{\Omega} \left(-z \frac{\partial v}{\partial y} + y \frac{\partial v}{\partial z} \right) d\Omega = 0 \quad \text{pour } v = \text{cste} \quad (22)$$

On vérifie cela numériquement en calculant $\sum_i f_i$ avant imposition de la condition aux limites, qui doit être proche de zéro.

Par ailleurs, la propriété d'antisymétrie constitue une vérification d'équilibre global : pour tout nœud (y_i, z_i) , on doit avoir $\omega(-y_i, z_i) + \omega(y_i, z_i) \approx 0$. Les résultats donnent une erreur maximale de $5.4 \times 10^{-4} \text{ m}^2$ sur le maillage 2D, soit **2.8%** de $|\omega|_{\max}$, ce qui est cohérent avec la précision d'un maillage de 316 nœuds.

6.3 Cohérence physique

Les trois vérifications suivantes confirment la cohérence physique des résultats :

1. **Symétries** : $\omega(-y, z) = -\omega(y, z)$ et $\omega(y, -z) = -\omega(y, z)$ sont satisfaites à moins de 3% sur le maillage grossier et à moins de 5% sur le maillage fin (erreur liée à l'asymétrie résiduelle du maillage).
2. **Valeurs aux coins** : $|\omega|_{\max} \approx 0.0196 \text{ m}^2$, proche de la limite théorique $|y_{\max} \times z_{\max}| = 0.05 \times 0.4375 = 0.0219 \text{ m}^2$. L'écart de 10% correspond aux corrections de la série de Saint-Venant.
3. **Contraintes** : τ_z varie linéairement en z et τ_y est concentrée aux extrémités — cohérent avec la théorie des plaques minces en torsion pour $L/e \gg 1$.

7 Conclusion

Ce travail a permis de construire et valider une chaîne de calcul éléments finis complète pour le problème de gauchissement de Saint-Venant. Les résultats obtenus sont en accord avec les prédictions théoriques :

- La fonction ω présente la forme en selle caractéristique et les antisymétries attendues,
- Les extrema aux coins ($\approx 0.0196 \text{ m}^2$) sont proches de la valeur limite $yz_{\max} = 0.02188 \text{ m}^2$, confirmant la cohérence physique de la solution,
- Les contraintes de cisaillement sont dominées par τ_z et cohérentes avec le fort élanement de la section.

Des améliorations envisageables incluent l'utilisation d'éléments $\mathcal{P}_2$ (6 nœuds) pour accélérer la convergence, ou l'application d'un lissage SPR (*Superconvergent Patch Recovery*) pour améliorer la régularité inter-éléments des contraintes.

A Code source — TP_elas.f90

Le code est présenté en cinq blocs distincts correspondant aux étapes du solveur.

A.1 — Déclarations et initialisation

Listing 2: Déclarations et lecture du maillage

```

1  program TP_elas
2      use mesh
3      implicit none
4
5      integer  :: i, r, c
6      integer  :: n1, n2, n3
7      integer  :: i_center
8      real(8)  :: ybar, zbar, dist, dist_min
9      real(8)  :: y1, y2, y3, z1, z2, z3, selem, y_c, z_c
10     real(8)  :: gradN(2,3), M_mat(3,3), M_inv(3,3)
11     real(8)  :: B_mat(2,3), Kel(3,3), qel(3)
12     real(8), allocatable :: Rsys(:,,:), vecs(:,,:), dep(:,,:), coor(:,,:)
13
14     nord = 1
15     call rw_mesh      ! lecture du fichier MAIL.msh
16
17     allocate(Rsys(nno,nno), vecs(nno,1), dep(nno,1), coor(2,nno))
18     Rsys = 0.0d0 ; vecs = 0.0d0 ; dep = 0.0d0
19
20     ! Operateur gradient de reference : extrait dN/dy et dN/dz
21     gradN      = 0.0d0
22     gradN(1,2) = 1.0d0      ! ligne 1 -> d/dy
23     gradN(2,3) = 1.0d0      ! ligne 2 -> d/dz

```

A.2 — Nœud central et coordonnées recentrées

Listing 3: Recherche du nœud central et recentrage

```

1      ! Centroïde geometrique du maillage
2      ybar = sum(ym) / nno
3      zbar = sum(zm) / nno
4
5      ! Nœud le plus proche du centroïde -> impose omega = 0
6      i_center = 1
7      dist_min = (ym(1)-ybar)**2 + (zm(1)-zbar)**2
8      do i = 2, nno
9          dist = (ym(i)-ybar)**2 + (zm(i)-zbar)**2
10         if (dist < dist_min) then
11             dist_min = dist
12             i_center = i
13         end if
14     end do
15
16     ! Coordonnees recentrees (pour l'export Gmsh)
17     coor(1,:) = ym - ybar
18     coor(2,:) = zm - zbar

```

A.3 — Boucle d'assemblage

Listing 4: Assemblage de la matrice de rigidité et du vecteur de charge

```

1  do i = 1, nelem
2      n1 = nodelem(1,i) ; n2 = nodelem(2,i) ; n3 = nodelem(3,i)
3      y1 = coor(1,n1) ; z1 = coor(2,n1)
4      y2 = coor(1,n2) ; z2 = coor(2,n2)
5      y3 = coor(1,n3) ; z3 = coor(2,n3)
6
7      ! Matrice de geometrie M et son inverse
8      M_mat(1,:) = [1.0d0, y1, z1]
9      M_mat(2,:) = [1.0d0, y2, z2]
10     M_mat(3,:) = [1.0d0, y3, z3]
11     call invermat(3, M_mat, M_inv)
12
13     ! B = gradN * M^{-1} (gradient constant sur l'element)
14     B_mat = matmul(gradN, M_inv)
15
16     ! Aire de l'element
17     selem = 0.5d0 * abs((y2-y1)*(z3-z1) - (y3-y1)*(z2-z1))
18
19     ! Matrice de rigidite et vecteur de charge elementaires
20     Kel = selem * matmul(transpose(B_mat), B_mat)
21     y_c = (y1+y2+y3)/3.0d0 ; z_c = (z1+z2+z3)/3.0d0
22     qel = selem * matmul(transpose(B_mat), [-z_c, y_c])
23
24     ! Assemblage global
25     do r = 1, 3
26         do c = 1, 3
27             Rsys(nodelem(r,i), nodelem(c,i)) = &
28             Rsys(nodelem(r,i), nodelem(c,i)) + Kel(r,c)
29         end do
30         vecs(nodelem(r,i), 1) = vecs(nodelem(r,i), 1) + qel(r)
31     end do
32 end do

```

A.4 — Condition aux limites et résolution

Listing 5: Imposition de la CL et résolution du système linéaire

```

1  ! Condition aux limites : omega = 0 au noeud central
2  ! (leve l'indetermination du probleme de Neumann pur)
3  Rsys(i_center, :) = 0.0d0
4  Rsys(i_center, i_center) = 1.0d0
5  vecs(i_center, 1) = 0.0d0
6
7  ! Resolution : K * dep = vecs (factorisation LU via LAPACK)
8  call linsys(nno, 1, Rsys, vecs, dep)
9
10 ! Export au format Gmsh et affichage
11 call figmshno_0(nno, nelem, nodelem, coor, dep(:,1))
12 write(*,*) 'Done. nno=', nno, ' nelem=', nelem
13 write(*,*) 'omega_min=', minval(dep), ' max=', maxval(dep)

```

A.5 — Sous-routines mathématiques

Listing 6: Inversion de matrice et solveur linéaire (LAPACK)


```
22      ! Donnees nodales (omega)
23      write(1,'(a)') '$NodeData' ; write(1,'(a)') '1'
24      write(1,'(a)') '"omega"' ; write(1,'(a)') '1'
25      write(1,'(a)') '□0.0' ; write(1,'(a)') '3'
26      write(1,'(a)') '0' ; write(1,'(a)') '1'
27      write(1,'(I6)') nno
28      do i = 1, nno
29          write(1,'(I6,1X,F20.10)') i, dep(i)
30      end do
31      write(1,'(a)') '$EndNodeData'
32      close(1)
33  end subroutine
34
35 end program TP_elas
```